(set-info :status unsat)
(set-option :produce-unsat-cores true)
(set-option :produce-unsat-assumptions true)
(set-logic QF_AUFBV)
(declare-fun a () (Array (_ BitVec 3) (_ BitVec 3)))
(declare-fun b () (Array (_ BitVec 3) (_ BitVec 3)))
(define-fun b0 () Bool (= (select a (_ bv0 3)) (select b (_ bv0 3))))
(define-fun b1 () Bool (= (select a (_ bv2 3)) (select b (_ bv2 3))))
(define-fun b2 () Bool (= (select a (_ bv4 3)) (select b (_ bv4 3))))
(define-fun b3 () Bool (= (select a (_ bv6 3)) (select b (_ bv6 3))))
(assert (! (= (select a (_ bv1 3)) (select b (_ bv1 3))) :named a0))
(assert (! (= (select a (_ bv3 3)) (select b (_ bv3 3))) :named a1))
(assert (! (= (select a (_ bv5 3)) (select b (_ bv5 3))) :named a2))
(assert (! (= (select a (_ bv7 3)) (select b (_ bv7 3))) :named a3))
(assert (! (not (= a b)) :named a4))
(check-sat-assuming (b0 b1 b2 b3))
(get-unsat-core)
(get-unsat-assumptions)
